

Quick Start Guide

iSeries Automated Document Distribution



Emailing Documents Automatically Using Report Splitter, Batch Report Server/400, and WebDocs iSeries Edition.

This guide highlights the process of using Report Splitter, Batch Report Server/400, and WebDocs iSeries Edition to automatically distribute your documents from WebDocs via email.

This allows you to:

- Control the WebDocs Document Type you want distributed.
- Programmatically determine the recipients.
- Decide if the recipient receives the documents as an attachment or a WebDocs URL.

This guide is divided into the following:

[Overview of Automated Document Distribution](#)

[Configuring Automated Document Distribution](#)

[Sample DOCEXITC Exit Program](#)

Overview of Automated Document Distribution

Automating document distribution processes saves time by automatically controlling how and to whom documents are distributed. The information below illustrates the path of a document, from spool file to distribution via email, and how each program along the path allows control over this process.

1. Report Splitter

Extracts important data from spool files and automatically splits them.



This iSeries product automates splitting a single spool file into multiple files using specific criteria, such as customer number, to control how they are split. It also, either in conjunction with or independently from splitting, extracts important values from spool files. By storing the values in the spool file's attributes, you can later assign them as search key values when your documents are automatically checked into WebDocs.

2. Batch Report Server/400


Converts spool files to more usable PDF documents and checks them in to WebDocs.



This product monitors an iSeries output queue for spool files. When it detects new spool files, it converts them to PDF documents. It then checks in the PDF documents to WebDocs. During the check-in process, the program automatically looks for the folder structure, Document Type, and search key values in the spool file's user defined data and assigns them in WebDocs.

3. WebDocs iSeries Edition

Emails documents from WebDocs to the desired recipients.



The final step in automated document distribution determines the recipient and method of distribution. The PDF documents are emailed to recipients by programmatically determining the email address of the recipient. Further, the extreme versatility of RJS products allows the documents to be attached in the email or sent as a WebDocs URL. Because WebDocs credentials are required for the URL, this provides additional security for sensitive documents.

Configuring Automated Document Distribution

This topic explains how to configure each of three iSeries products to integrate them for automating document distribution. While a document moves from Report Splitter to Batch Report Server/400 to WebDocs iSeries Edition, you configure these products in a different order.

This guide shows two distinct ways for programmatically obtaining an email address:

Option 1: Extract the email address from the spool file.

Option 2: Query the database for an email address.

If these two options are not relevant to your system configuration, or you are not sure how your system is configured, contact [RJS Software sales](#) to discuss your options.

This topic is divided into the following:

[Configuring WebDocs iSeries Edition](#)

[Configuring Report Splitter](#)

[Configuring Batch Report Server/400](#)

Configuring WebDocs iSeries Edition

First, you need to configure WebDocs iSeries Edition. This guide assumes that the WebDocs folders, Document Types, and search keys, are already set up. For help planning your WebDocs configuration, contact [RJS Software sales](#) for the WebDocs Workbook; for help setting it up see the [WebDocs Administrator Guide](#).

Note: The configuration in this guide requires specific knowledge of exit programming with WebDocs iSeries Edition. If you are not familiar with exit programming, contact [RJS Software technical support](#) to find out more information on RJS Software's professional services.

This topic is divided into the following:

[Gathering the Required Information](#)

[Obtaining the Special Library](#)


[Modifying and Compiling the Exit Program](#)

iSeries Automated Document Distribution Quick Start Guide



Gathering the Required Information

You need to gather information about your WebDocs configuration before configuring iSeries automated document distribution. When the document reaches Batch Report Server/400 and is ready for WebDocs check-in, the values for folder structure, Document Type, and search keys must match the configuration in WebDocs.



The above graphic illustrates information you need from your WebDocs setup. Gathering this information is extremely important. In this example, you would gather the following information:

- Folder name and level: AR and folder level 1.
- Document Type: AR.
- Search Keys:
 1. Invoice Number
 2. Customer Number
 3. Customer Name
 4. Document Date

iSeries Automated Document Distribution Quick Start Guide



Obtaining the Special iSeries Library

Some operations in the exit program require a special library called **RJSMODS**. To obtain this library contact [RJS Software technical support](#).

Modifying and Compiling the Exit Program

You need to use a modified exit program in WebDocs iSeries Edition to automatically email specific types of documents at the post-check-in exit point. The source code for the exit program, **DOCEXITC**, is located in the **RJSIMAGE/SOURCE** library. Only modify a copy of the source file, not the original, to reflect your specific WebDocs configuration. The following example uses the same sample information from [Gathering the Required Information](#). You can find the full [sample exit program](#) at the end of this guide.

To modify a copy of the exit program source code to reflect your WebDocs configuration:

1. Copy the exit program source to your own library before modifying it.
2. In addition to the standard variable declarations, add new variables to temporarily store values used for email distribution:

- Variables for the recipient's email address and the IFS file path:

```
DCL          VAR(&EMAILTO) TYPE(*CHAR) LEN(256)
DCL          VAR(&IFSFILE) TYPE(*CHAR) LEN(255)
```

- Miscellaneous variables used throughout the program:

```
DCL          VAR(&RTNVAL) TYPE(*CHAR) LEN(512)
DCL          VAR(&EMAILTMP) TYPE(*CHAR) LEN(256)
DCL          VAR(&I) TYPE(*DEC) LEN(3 0)
DCL          VAR(&J) TYPE(*DEC) LEN(3 0)
```

3. Define which Document Type the exit program will process; this exit program will ignore documents with other Document Types:

```
IF          COND(&DOCTYPE2 *EQ 'AR') THEN(DO)
```

4. Obtain the recipient's email address:

Option 1: The email address has been extracted from the spool file, and stored in the user defined data attribute:

Load the user defined data into **&RTNVAL**:

```
RJSMODS/MODLOOKUP LOOKUPEXP(DCUSRDFN) TABLE('RJSIMAGE/DOCS00') +
WHERE('DOCID = @QT@P01@QT') RETURNVAR(&RTNVAL) SUBST(&DOCID)
```

Check that "mailto:" is present in the user defined data:

```
IF          COND(%SST(&RTNVAL 1 5) *NE 'ERROR') THEN(DO)
CHGVAR     VAR(&EMAILTO) VALUE(' ')
CHGVAR     VAR(&I) VALUE(1)
```

iSeries Automated Document Distribution Quick Start Guide



```
DOWHILE COND(%SST(&RTNVAL &I 2) *NE ' ' *AND &EMAILTO *EQ ' ')
```

If “mailto:” is found, load all proceeding characters into **&EMAILTMP**:

```
IF COND(%SST(&RTNVAL &I 7) *EQ 'MAILTO:') THEN(DO)
CHGVAR VAR(&I) VALUE(&I + 7)
CHGVAR VAR(&J) VALUE(256 - &I)
CHGVAR VAR(&EMAILTMP) VALUE(%SST(&RTNVAL &I &J))
CHGVAR VAR(&I) VALUE(1)
```

Loop through each character in **&EMAILTMP** to build the email address into **&EMAILTO**, stopping at the tilde (~):

```
DOWHILE COND(%SST(&EMAILTMP &I 1) *NE '~')
CHGVAR VAR(&EMAILTO) VALUE(&EMAILTO *TCAT +
%SST(&EMAILTMP &I 1))
CHGVAR VAR(&I) VALUE(&I + 1)
ENDDO
ENDDO
CHGVAR VAR(&I) VALUE(&I + 1)
ENDDO
```

Option 2: If the email address will be obtained by querying a database:

Assumptions: This sample code assumes that a table named **CUSTMAST** exists inside the library **YOURLIB** on your iSeries (**YOURLIB/CUSTMAST**). **CUSTMAST** is queried with customer number (**KEY2**) and returns the value of its **EMAILADDR** field.

```
IF COND(&EMAILTO *EQ ' ') THEN(DO)
RJSMODS/MODLOOKUP LOOKUPEXP(EMAILADDR) TABLE('YOURLIB/CUSTMAST') +
WHERE('CUSTNO = @P01') RETURNVAR(&RTNVAL) SUBST(&KEY2)
IF COND(%SST(&RTNVAL 1 5) *NE 'ERROR') THEN(DO)
CHGVAR VAR(&EMAILTO) VALUE(&RTNVAL)
ENDDO
ENDDO
```

5. Retrieve the file to be sent:

```
RJSMODS/MODIMGGET DOCID(&DOCID) IFSFILE(&IFSFILE)
```

6. Choose one of the following two customizable options for sending the document:

- If you want the document attached to the email:

```
DOCSEND2 TOADDR(&EMAILTO) FROMADDR(*DEFAULT) FROMNAME(*DEFAULT) +
SUBJECT('Invoice ' *CAT &KEY1) MESSAGE('Please find your +
invoice attached to this email.') ATTACHMENT(&IFSFILE) +
SMTPHOST(*DEFAULT)
```

- If you want to send a WebDocs URL for the document in the email:

```
DOCSENDLNK DOCID(&DOCID) TOADDR(&EMAILTO) FROMADDR(*DEFAULT) +
SUBJECT('WebDocs Link to Invoice ' *CAT &KEY1)
```

7. Compile the modified exit program source code to the **RJSIMAGE** library. WebDocs is now configured for automated document distribution.

iSeries Automated Document Distribution Quick Start Guide



Configuring Report Splitter

Second, you need to configure Report Splitter. Report Splitter processes spool files in an iSeries output queue by extracting values and splitting the spool file into multiple files. Report Splitter can also extract values for folder, Document Type, and search keys, and store them in the user defined data attribute of each new spool file. To learn more about Report Splitter, including many topics not covered in this guide, see the [Report Splitter User Guide](#).

To configure Report Splitter for automated document distribution:

1. From the Report Splitter main menu, choose **20. Edit Report Page Definitions**. Press **F6** to create a new input set-up page and enter the following:
 - a) **Spool:** Name of the spool files.
 - b) **Desc:** A meaningful description.
2. For each value extracted from the spool file enter the following:

Note: If you are using [option 2](#) to query a database for the recipient's email address, you do not need to include a value for email address as shown in the screenshot below.

- a) **Line:** The line on which this data is located.
- b) **Beg Col:** Beginning column for this data.
- c) **End Col:** Ending column for this data.
- d) **Desc:** A meaningful description.
- e) **Symbolic Field Name:** Variable name with a leading @ symbol.

```
AS/400 Report Splitter
Edit Report Page Definition

Create
Spool: INVOICE Type: *DEFAULT Enabled: Y
Desc: All invoices

Symbolic Field Name
(Always Start with @) Fill Activated
Line Beg/End Col Desc @INNUM N Y
8 2 6 Invoice Number @CUSTNUM N Y
8 32 35 Customer Number @CUSTNAME N Y
8 37 52 Customer Name @MAILTO N Y
8 54 83 Email Address @DATE N Y
3 23 30 Document Date @FLD6 N Y
@FLD7 N Y
```

3. Press **Enter** to return to the Report Splitter main menu.

iSeries Automated Document Distribution Quick Start Guide



4. Choose **21. Edit Report Distribution Definitions**. Press **F6** to create a new output set-up page and enter the following:
 - a) **Spool:** Name of the spool files.
 - b) **Entry Description:** A meaningful description.
 - c) **Dest Outq/Library:** The output queue name and library to hold processed spool files.
 - d) **Split Option:** Option to split spool files.

For example: If the spool file is a batch invoice and uses the split value of invoice number from [step 2](#), use ***DECOLLATE** as the split option. If the spool file is an individual invoice and does not need to be split, use ***NOSPLIT**.

```
AS/400 Report Splitter
Edit Report Distribution Settings
Create

Spool File: INVOICE Group: *DEFAULT Sequence: 0001464 Activate: Y
Entry Description: Extract WebDocs search keys and split by invoice
Dest Outq/Library: INVOICES / QGPL User/Addr:
Split Option: *DECOLLATE (Press F11 to Enter/View Report Selection Criteria)
Page Range: - Comp Message: N Audit Trail: N
Use Extended Select: Y Pgm: *ANY Job: Form:
```

5. Press **F13** and, in **New User Defined Data**, use the format below to input values in the user defined data field of the spool file.

For example: using the same sample information from [Gathering the Required Information](#), you would enter the following:

Option 1: If you are extracting the email address from the spool file:


```
IMGSERVER:@SPOOL;@INVNUM;@CUSTNUM;@CUSTNAME;@DATE; ; ; ; ; ;
;AR;AR Folder; ; ; ; ~MAILTO:@MAILTO;~
```

Option 2: When a database is queried to obtain an email address:

```
IMGSERVER:@SPOOL;@INVNUM;@CUSTNUM;@CUSTNAME;@DATE; ; ; ; ; ;
;AR;AR; ; ; ; ; ~
```


Note: Type a space when you are not using values.

iSeries Automated Document Distribution Quick Start Guide



6. Press **Enter** to return to the Report Splitter main menu.
7. Choose **11. Run Report Splitter Output Queue Monitor – RPTMON** and enter the following:
 - a) **Source output queue:** The output queue to monitor.
 - b) **Library:** The library for this output queue.
 - c) **After processing spool option:** Action taken after input spool file is processed.
 - d) **Delay processing or run once:** for production systems, use ***DELAY**.
 - e) **Move to output queue:** When **After Processing Option** is set to ***MOVE**, this is the output queue where the input spool files are moved.
 - f) **Library:** Library for the output queue containing moved input spool files.

Note: Moving processed spool files to an archive is optional, and may be beneficial should you ever need to access the original spool files again.



8. Press **Enter** and Report Splitter is configured.

Configuring Batch Report Server/400

Finally, configure Batch Report Server/400. This program acts as a link between processed spool files and WebDocs. Batch Report Server/400 monitors the output queue where Report Splitter sends output spool files. Batch Report Server/400 finds spool files in this output queue, converts them to PDF and checks them in to WebDocs. To learn more about Batch Report Server/400, including many topics not covered in this guide, see the [Batch Report Server/400 Administrator Guide](#).

To configure Batch Report Server/400 for automated document distribution:

1. From the Batch Report Server/400 main menu, choose **3. Edit Output Queue List**. Press **F6** to create a new output queue monitor page and enter the following:
 - a. **Src Output Queue Name:** The output queue where Report Splitter sends processed spool files.
 - b. **Src Output Queue Library:** The library for this output queue.
 - c. **IFS Dir:** Temporary directory used for conversion, enter */rjstemp*.
 - d. **Output Data Type:** Leave as PDF.
 - e. **After Processing:** **DELETE* to delete spool files from the output queue after converting them to PDF documents and checking them in to WebDocs.

```
Batch Report Server/400
Edit User Output Queue Settings
Create
Sequence . . . . : 377      Output Queue Group: _____
Src Output Queue Name: INVOICES  Src Output Queue Library: QGPL
IFS Dir: /rjstemp
Output Data Type: PDF          Move to Output Queue: _____
Activate Entry: Y (Y/N)       Move to Library: _____
After Processing: *DELETE     Ext Spool File: _____
Print Control Chr: *PRTCTL    Ext Spool Userdata: _____
Use Extended Select: N (Y/N)  Ext Spool Userid: _____
Ext Pgm Crtd Rpt: _____  Ext Spool Job: _____
```

2. Press **Enter**. On the next screen enter the following:

Note: Do not change the values for Doc Type and Folder because Report Splitter has already stored them in the spool file's user defined data attribute.

- a. **Send to WebDocs:** *Y* to check these documents in to WebDocs.
- b. **WebDocs Checkin User:** The WebDocs user to use when checking in documents. This WebDocs user must have access and check-in permissions to the folder.

iSeries Automated Document Distribution Quick Start Guide



```
Batch Report Server/400
Edit User Output Queue Settings
Create

Sequence . . . . : 377
Send to WebDocs: Y WebDocs Checkin User: ADMIN
Doc Type: _____
WebDocs Folder:
Folder 1: TEST
```

3. Press **Enter** and Batch Report Server/400 is configured. This completes the configuration of the three iSeries products needed for automating document distribution via email.

iSeries Automated Document Distribution Quick Start Guide



Sample DOCEXITC Exit Program

```
/* **** */
/*      Author:   Amanda Smith                               */
/*      Date written: 10/31/2012                             */
/*      Purpose:   Automated Doc Distribution (iSeries Email) */
/*Expected parameters: None                                  */
/*      Switches used: None                                   */
/*      LDA:      None                                       */
/*      Version:   3.21.00                                    */
/*      Desc:      Email Document Post Check-in              */
/* **** */
PGM      PARM(&DOCID &REVISION &DOCTITLE &DOCFLR1 &DOCFLR2 &DOCFLR3 +
            &DOCFLR4 &DOCFLR5 &KEY1 &KEY2 &KEY3 &KEY4 &KEY5 &KEY6 &KEY7 &KEY8 +
            &KEY9 &KEY10 &DOCPATH &DOCFILE &DOCTYPE &DOCTYPE2)

DCL      VAR(&DOCID) TYPE(*CHAR) LEN(100)
DCL      VAR(&REVISION) TYPE(*DEC) LEN(9 0)
DCL      VAR(&DOCTITLE) TYPE(*CHAR) LEN(200)
DCL      VAR(&DOCFLR1) TYPE(*CHAR) LEN(100)
DCL      VAR(&DOCFLR2) TYPE(*CHAR) LEN(100)
DCL      VAR(&DOCFLR3) TYPE(*CHAR) LEN(100)
DCL      VAR(&DOCFLR4) TYPE(*CHAR) LEN(100)
DCL      VAR(&DOCFLR5) TYPE(*CHAR) LEN(100)
DCL      VAR(&KEY1) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY2) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY3) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY4) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY5) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY6) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY7) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY8) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY9) TYPE(*CHAR) LEN(200)
DCL      VAR(&KEY10) TYPE(*CHAR) LEN(200)
DCL      VAR(&DOCPATH) TYPE(*CHAR) LEN(255)
DCL      VAR(&DOCFILE) TYPE(*CHAR) LEN(255)
DCL      VAR(&DOCTYPE) TYPE(*CHAR) LEN(10)
DCL      VAR(&DOCTYPE2) TYPE(*CHAR) LEN(20) /* 1.26 */
/* New variables for email distribution */
DCL      VAR(&EMAILTMP) TYPE(*CHAR) LEN(256)
DCL      VAR(&EMAILTO) TYPE(*CHAR) LEN(256)
DCL      VAR(&IFSFILE) TYPE(*CHAR) LEN(255)
DCL      VAR(&RTNVAL) TYPE(*CHAR) LEN(512)
DCL      VAR(&I) TYPE(*DEC) LEN(3 0)
DCL      VAR(&J) TYPE(*DEC) LEN(3 0)

/* Data areas set by DOCCHKIN command */
DCL      VAR(&DOCIDTMP) TYPE(*CHAR) LEN(100) /* 3.21.00 */
DCL      VAR(&DOCUSERTMP) TYPE(*CHAR) LEN(20) /* 3.21.00 */
DCL      VAR(&DOCFILETMP) TYPE(*CHAR) LEN(255) /* 3.21.00 */

MONMSG      MSGID(CPF0000) EXEC(GOTO CMDLBL(ERRORS))

/* Retrieve data area values from DOCCHKIN */
RTVDTAARA DTAARA(DOCIDTMP) RTNVAR(&DOCIDTMP) /* 3.21.00 */
```

iSeries Automated Document Distribution

Quick Start Guide



```
MONMSG      MSGID(CPF0000)
RTVDTAARA   DTAARA(DOCUSERTMP) RTNVAR(&DOCUSERTMP) /* 3.21.00 */
MONMSG      MSGID(CPF0000)
RTVDTAARA   DTAARA(DOCFILETMP) RTNVAR(&DOCFILETMP) /* 3.21.00 */
MONMSG      MSGID(CPF0000)

/* Do whatever processing is required. If possible, always do additional +
   processing via the SBMJOB command to prevent the web server from +
   locking up. Example: SBMJOB  JOB(IMGCHECKIN) */

/* If you use files in your own libraries, be sure to use the +
   ADDLIBLE command to ensure that those files are available */

/* Email customer the document type invoice */
IF          COND(&DOCTYPE2 *EQ 'INVOICE') THEN(DO)
  /* For this document type, KEY1 is invoice number, +
     and KEY2 is the customer number. */

  /* First check to see if there is an email address if this +
     document came from a spool file. */
  RJSMODS/MODLOOKUP LOOKUPEXP(DCUSRDFN) TABLE('RJSIMAGE/DOCS00') +
     WHERE('DOCID = @QT@P01@QT') RETURNVAR(&RTNVAL) SUBST(&DOCID)
  IF          COND(%SST(&RTNVAL 1 5) *NE 'ERROR') THEN(DO)
    CHGVAR    VAR(&EMAILTO) VALUE(' ')
    CHGVAR    VAR(&I) VALUE(1)

    /* If we hit two spaces or find email we know we're done */
    DOWHILE   COND(%SST(&RTNVAL &I 2) *NE ' ' *AND &EMAILTO *EQ ' ')

    IF          COND(%SST(&RTNVAL &I 7) *EQ 'MAILTO:') THEN(DO)

      /* &I is the beginning position of the email address */
      CHGVAR    VAR(&I) VALUE(&I + 7)

      /* &J is the remaining length of the user defined data */
      CHGVAR    VAR(&J) VALUE(256 - &I)

      /* Put the email address into a temporary variable +
         so we can look for the terminating character. */
      CHGVAR    VAR(&EMAILTMP) VALUE(%SST(&RTNVAL &I &J))

      /* Reset &I */
      CHGVAR    VAR(&I) VALUE(1)

      /* Loop through the email address until we find ~ */
      DOWHILE   COND(%SST(&EMAILTMP &I 1) *NE '~')

        /* Build up our actual email address one character +
           at a time until we find the terminating ~ */
        CHGVAR    VAR(&EMAILTO) VALUE(&EMAILTO *TCAT +
           %SST(&EMAILTMP &I 1))
        CHGVAR    VAR(&I) VALUE(&I + 1)
      ENDDO

    ENDDO

  ENDDO

  /* &RTNVAL has MAILTO: */
```

iSeries Automated Document Distribution Quick Start Guide



```
        /* Keep looking if we haven't found email or +
           run out of text. */
        CHGVAR      VAR(&I) VALUE(&I + 1)

        ENDDO      /* &RTNVAL not blank and no email */

        ENDDO      /* &RTNVAL was not an error */

        /* If we didn't get the email address from a spool file, +
           then we'll need to look it up in the customer master file. */
        IF          COND(&EMAILTO *EQ ' ') THEN(DO)
        /* The assumption is that file CUSTMAST has a field named EMAILADDR +
           that is the email address we want, and that the customer number +
           field is numeric and named CUSTNO. */
        RJSMODS/MODLOOKUP LOOKUPEXP(EMAILADDR) TABLE('YOUURLIB/CUSTMAST') +
           WHERE('CUSTNO = @P01') RETURNVAR(&RTNVAL) SUBST(&KEY2)
        IF          COND(%SST(&RTNVAL 1 5) *NE 'ERROR') THEN(DO)
        CHGVAR      VAR(&EMAILTO) VALUE (&RTNVAL)
        ENDDO
        ENDDO

        /* For simplicity, we'll assume that we have an email +
           address at this point */

        /* Get the file name for this document */
        RJSMODS/MODIMGGET DOCID(&DOCID) IFSFILE(&IFSFILE)

        /* Send the invoice to the customer as an attachment */
        DOCSEND2    TOADDR(&EMAILTO) FROMADDR(*DEFAULT) FROMNAME(*DEFAULT) +
           SUBJECT('Invoice ' *CAT &KEY1) MESSAGE('Please find your +
           invoice attached to this email.') ATTACHMENT(&IFSFILE) +
           SMTPHOST(*DEFAULT)

        /* Or, send the invoice to the customer as a link */
        DOCSENDLNK DOCID(&DOCID) TOADDR(&EMAILTO) FROMADDR(*DEFAULT) +
           SUBJECT('WebDocs Link to Invoice ' *CAT &KEY1)

        ENDDO      /* Doc type is INVOICE */

        RETURN     /* Normal exit */

        /*****/
        /* Handle errors */
        /*****/
        ERRORS:
        SNDPGMMSG  MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA('Errors occurred while +
           running WebDocs document check-in for document ID:' |> &DOCID) +
           MSGTYPE(*INFO)
        MONMSG     MSGID(CPF0000) /* Make sure we exit */
        SNDPGMMSG  MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA('Errors occurred while +
           running WebDocs document check-in for document ID:' |> &DOCID) +
           TOMSGQ(*SYSOPR)
        MONMSG     MSGID(CPF0000) /* Make sure we exit */

        ENDPGM
```